
Factorial ANOVA 
/* potato.sas */ 
options linesize=79 noovp formdlim='_'; 
title 'Rotten potatoes'; 
title2 'Two-factor ANOVA several different ways'; 
 
proc format; 
     value tfmt 1 = 'Cool'  2 = 'Warm'; 
 
data spud; 
     infile 'potato2.data' firstobs=2; /* Skip the first line that R uses */ 
     input id bact temp rot; 
     /* Cell means coding for all 6 treatment combinations */ 
     if temp=1 and bact=1 then mu11=1; else mu11=0; 
     if temp=1 and bact=2 then mu12=1; else mu12=0; 
     if temp=1 and bact=3 then mu13=1; else mu13=0; 
     if temp=2 and bact=1 then mu21=1; else mu21=0; 
     if temp=2 and bact=2 then mu22=1; else mu22=0; 
     if temp=2 and bact=3 then mu23=1; else mu23=0; 
     combo = 10*temp+bact; 
     format temp tfmt.; 
 
proc means; 
     class bact temp; 
     var rot; 
 
/* Better looking output from proc tabulate */ 
 
proc tabulate; 
     class bact temp; 
     var rot; 
     table (temp all),(bact all) * (mean*rot); 
 
proc glm; 
     title3 'Standard 2-way ANOVA with proc glm'; 
     class bact temp; 
     model rot=temp|bact; /* Could have said   bact temp bact*temp */ 
     means temp|bact; 
 
/* Need to plot it; SAS is not the tool. */ 

/* Now generate the tests for main effects and interaction using cell means 
   coding.  
 
               BACTERIA TYPE 
TEMP      1         2        3 
Cool    mu11      mu12      mu13 
Warm    mu21      mu22      mu23                      */ 
 
 
/* The test statement of proc reg uses variable names to stand for the 
corresponding regression coefficients. By naming the effect cell mean coding 
dummy variables the same as the population cell means, I can just state the 
null hypothesis. Isn't this a cute SAS trick? */ 
 
proc reg; 
     title3 'Using the proc reg test statement and cell means coding'; 
     model rot = mu11--mu23 / noint; 
     Overall:       test mu11=mu12=mu13=mu21=mu22=mu23; 
     Temperature:   test mu11+mu12+mu13 = mu21+mu22+mu23; 
     Bacteria:      test mu11+mu21 = mu12+mu22 = mu13+mu23; 
     Bact_by_Temp1: test mu11-mu21 = mu12-mu22 = mu13-mu23; 
     Bact_by_Temp2: test mu12-mu11 = mu22-mu21, 
                         mu13-mu12 = mu23-mu22; 
 
/* Bact_by_Temp1 checks equality of temperature effects.  
   Bact_by_Temp2 checks parallel line segments. They are equivalent. */ 
 
proc glm; 
     title3 'Proc glm: Using contrasts on the combination variable'; 
     class combo;      /* 11 12 13  21 22 23 */ 
     model rot=combo; 
     contrast 'Main Effect for Temperature' 
        combo  1  1  1  -1 -1 -1; 
     contrast 'Main Effect for Bacteria' 
        combo  1 -1  0   1 -1  0, 
        combo  0  1 -1   0  1 -1; 
     contrast 'Temperature by Bacteria Interaction' 
        combo  1 -1  0  -1  1  0, 
        combo  0  1 -1   0 -1  1; 
 
/* Illustrate effect coding */ 
 
data mashed; 
     set spud; 
     /* Effect coding, with interactions */ 
        if bact = 1 then b1 =  1; 
           else if bact = 2 then b1 =  0; 
           else if bact = 3 then b1 = -1; 
        if bact = 1 then b2 =  0; 
           else if bact = 2 then b2 =  1; 
           else if bact = 3 then b2 = -1; 
        if temp = 1 then t =  1; 
           else if temp = 2 then t = -1; 
     /* Interaction terms */ 
        tb1 = t*b1; tb2 = t*b2; 

proc reg; 
     title3 'Effect coding'; 
     model rot = b1 b2  t  tb1 tb2; 
     Temperature:   test t=0; 
     Bacteria:      test b1=b2=0; 
     Bact_by_Temp:  test tb1=tb2=0; 
 
/* Do some exploration to follow up the interaction. The regression 
   with cell means coding is easiest. The final product of several runs 
   is shown below. For reference, here is the table of population means again. 
 
               BACTERIA TYPE 
TEMP      1         2        3 
Cool    mu11      mu12      mu13 
Warm    mu21      mu22      mu23                      */ 
 
proc reg; 
     title3 'Further exploration using cell means coding'; 
     model rot = mu11--mu23 / noint; 
     /* Pairwise comparisons of marginal means for Bacteria Type */ 
     Bact1vs2: test mu11+mu21=mu12+mu22; 
     Bact1vs3: test mu11+mu21=mu13+mu23; 
     Bact2vs3: test mu12+mu22=mu13+mu23; 
     /* Effect of temperature for each bacteria type */ 
     Temp_for_Bac1: test mu11=mu21; 
     Temp_for_Bac2: test mu12=mu22; 
     Temp_for_Bac3: test mu13=mu23; 
     /* Effect of bacteria type for each temperature */ 
     Bact_for_CoolTemp: test mu11=mu12=mu13; 
     Bact_for_WarmTemp: test mu21=mu22=mu23; 
     /* Pairwise comparisons of bacteria types at warm temperature */ 
     Bact1vs2_for_WarmTemp: test mu21=mu22; 
     Bact1vs3_for_WarmTemp: test mu21=mu23; 
     Bact2vs3_for_WarmTemp: test mu22=mu23; 
 
/* We have done a lot of tests. Concerned about buildup of Type I 
error? We can make ALL the tests into Scheffe follow-ups of the 
initial test for equality of the 6 cell means. The Scheffe family 
includes all COLLECTIONS of contrasts, not just all contrasts. */ 
 
proc iml; 
     title3 'Table of critical values for all possible Scheffe tests'; 
     numdf = 5;   /* Numerator degrees of freedom for initial test */ 
     dendf =48;   /* Denominator degrees of freedom for initial test */ 
     alpha = 0.05; 
     critval = finv(1-alpha,numdf,dendf);  
     zero = {0 0}; S_table = repeat(zero,numdf,1);  /* Make empty matrix */ 
     /* Label the columns */ 
     namz = {"Number of Contrasts in followup test"  
             "    Scheffe Critical Value"}; mattrib S_table colname=namz; 
     do i = 1 to numdf; 
        s_table(|i,1|) = i;  
        s_table(|i,2|) = numdf/i *  critval; 
     end; 
     reset noname; /* Makes output look nicer in this case */ 
     print "Initial test has"  numdf " and " dendf "degrees of freedom." 
           "Using significance level alpha = " alpha; 
     print s_table; 

_______________________________________________________________________________ 
  
                                Rotten potatoes                               1 
                    Two-factor ANOVA several different ways 
 
                              The MEANS Procedure 
 
                            Analysis Variable : rot  
  
                          N 
          bact   temp   Obs    N           Mean        Std Dev        Minimum 
  --------------------------------------------------------------------------- 
             1   Cool     9    9      3.5555556      4.2752518              0 
 
                 Warm     9    9      7.0000000      3.5355339              0 
 
             2   Cool     9    9      4.7777778      3.1135903              0 
 
                 Warm     9    9     13.5555556      6.3267510      3.0000000 
 
             3   Cool     9    9      8.0000000      4.5552168      2.0000000 
 
                 Warm     9    9     19.5555556      5.5251948      8.0000000 
  --------------------------------------------------------------------------- 
 
                           Analysis Variable : rot  
  
                                           N 
                           bact   temp   Obs        Maximum 
                   ---------------------------------------- 
                              1   Cool     9      9.0000000 
 
                                  Warm     9     10.0000000 
 
                              2   Cool     9     10.0000000 
 
                                  Warm     9     23.0000000 
 
                              3   Cool     9     15.0000000 
 
                                  Warm     9     26.0000000 
                   ---------------------------------------- 

_______________________________________________________________________________ 
  
                                Rotten potatoes                               2 
                    Two-factor ANOVA several different ways 
 
    ----------------------------------------------------------------------- 
    |                 |                 bact                 |            | 
    |                 |--------------------------------------|            | 
    |                 |     1      |     2      |     3      |    All     | 
    |                 |------------+------------+------------+------------| 
    |                 |    Mean    |    Mean    |    Mean    |    Mean    | 
    |                 |------------+------------+------------+------------| 
    |                 |    rot     |    rot     |    rot     |    rot     | 
    |-----------------+------------+------------+------------+------------| 
    |temp             |            |            |            |            | 
    |-----------------|            |            |            |            | 
    |Cool             |        3.56|        4.78|        8.00|        5.44| 
    |-----------------+------------+------------+------------+------------| 
    |Warm             |        7.00|       13.56|       19.56|       13.37| 
    |-----------------+------------+------------+------------+------------| 
    |All              |        5.28|        9.17|       13.78|        9.41| 
    ----------------------------------------------------------------------- 
  
_______________________________________________________________________________ 
  
                                Rotten potatoes                               3 
                    Two-factor ANOVA several different ways 
                      Standard 2-way ANOVA with proc glm 
 
                               The GLM Procedure 
 
                           Class Level Information 
  
                      Class         Levels    Values 
 
                      bact               3    1 2 3      
 
                      temp               2    Cool Warm  
 
 
                    Number of Observations Read          54 
                    Number of Observations Used          54 

_______________________________________________________________________________ 
  
                                Rotten potatoes                               4 
                    Two-factor ANOVA several different ways 
                      Standard 2-way ANOVA with proc glm 
 
                               The GLM Procedure 
  
Dependent Variable: rot    
 
                                      Sum of 
Source                     DF        Squares    Mean Square   F Value   Pr > F 
 
Model                       5    1652.814815     330.562963     15.05   <.0001 
 
Error                      48    1054.222222      21.962963                    
 
Corrected Total            53    2707.037037                                   
 
 
              R-Square     Coeff Var      Root MSE      rot Mean 
 
              0.610562      49.81676      4.686466      9.407407 
 
 
Source                     DF      Type I SS    Mean Square   F Value   Pr > F 
 
temp                        1    848.0740741    848.0740741     38.61   <.0001 
bact                        2    651.8148148    325.9074074     14.84   <.0001 
bact*temp                   2    152.9259259     76.4629630      3.48   0.0387 
 
 
Source                     DF    Type III SS    Mean Square   F Value   Pr > F 
 
temp                        1    848.0740741    848.0740741     38.61   <.0001 
bact                        2    651.8148148    325.9074074     14.84   <.0001 
bact*temp                   2    152.9259259     76.4629630      3.48   0.0387 

_______________________________________________________________________________ 
  
                                Rotten potatoes                               5 
                    Two-factor ANOVA several different ways 
                      Standard 2-way ANOVA with proc glm 
 
                               The GLM Procedure 
 
               Level of            -------------rot------------- 
               temp          N             Mean          Std Dev 
 
               Cool         27        5.4444444       4.31752541 
               Warm         27       13.3703704       7.27031979 
 
 
               Level of            -------------rot------------- 
               bact          N             Mean          Std Dev 
 
               1            18        5.2777778       4.19811660 
               2            18        9.1666667       6.61771242 
               3            18       13.7777778       7.71214135 
 
 
         Level of     Level of           -------------rot------------- 
         bact         temp         N             Mean          Std Dev 
 
         1            Cool         9        3.5555556       4.27525178 
         1            Warm         9        7.0000000       3.53553391 
         2            Cool         9        4.7777778       3.11359028 
         2            Warm         9       13.5555556       6.32675097 
         3            Cool         9        8.0000000       4.55521679 
         3            Warm         9       19.5555556       5.52519482 

_______________________________________________________________________________ 
  
                                Rotten potatoes                               6 
                    Two-factor ANOVA several different ways 
            Using the proc reg test statement and cell means coding 
 
                               The REG Procedure 
                                 Model: MODEL1 
                           Dependent Variable: rot  
 
                    Number of Observations Read          54 
                    Number of Observations Used          54 
 
 
              NOTE: No intercept in model. R-Square is redefined. 
 
                             Analysis of Variance 
  
                                    Sum of           Mean 
Source                   DF        Squares         Square    F Value    Pr > F 
 
Model                     6     6431.77778     1071.96296      48.81    <.0001 
Error                    48     1054.22222       21.96296                      
Uncorrected Total        54     7486.00000                                     
 
 
             Root MSE              4.68647    R-Square     0.8592 
             Dependent Mean        9.40741    Adj R-Sq     0.8416 
             Coeff Var            49.81676                        
 
 
                             Parameter Estimates 
  
                          Parameter       Standard 
     Variable     DF       Estimate          Error    t Value    Pr > |t| 
 
     mu11          1        3.55556        1.56216       2.28      0.0273 
     mu12          1        4.77778        1.56216       3.06      0.0036 
     mu13          1        8.00000        1.56216       5.12      <.0001 
     mu21          1        7.00000        1.56216       4.48      <.0001 
     mu22          1       13.55556        1.56216       8.68      <.0001 
     mu23          1       19.55556        1.56216      12.52      <.0001 
  
_______________________________________________________________________________ 

                                Rotten potatoes                               7 
                    Two-factor ANOVA several different ways 
            Using the proc reg test statement and cell means coding 
 
                               The REG Procedure 
                                 Model: MODEL1 
 
                Test Overall Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           5      330.56296      15.05    <.0001 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 
 
              Test Temperature Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      848.07407      38.61    <.0001 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 
 
               Test Bacteria Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2      325.90741      14.84    <.0001 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 
 
             Test Bact_by_Temp1 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2       76.46296       3.48    0.0387 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 
 
 
             Test Bact_by_Temp2 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2       76.46296       3.48    0.0387 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 

                                Rotten potatoes                              12 
                    Two-factor ANOVA several different ways 
             Proc glm: Using contrasts on the combination variable 
 
                               The GLM Procedure 
 
                           Class Level Information 
  
                  Class         Levels    Values 
 
                  combo              6    11 12 13 21 22 23  
 
 
                    Number of Observations Read          54 
                    Number of Observations Used          54 
  
_______________________________________________________________________________ 
 
Dependent Variable: rot    
 
                                      Sum of 
Source                     DF        Squares    Mean Square   F Value   Pr > F 
 
Model                       5    1652.814815     330.562963     15.05   <.0001 
Error                      48    1054.222222      21.962963                    
Corrected Total            53    2707.037037                                   
 
 
              R-Square     Coeff Var      Root MSE      rot Mean 
 
              0.610562      49.81676      4.686466      9.407407 
 
 
Source                     DF      Type I SS    Mean Square   F Value   Pr > F 
 
combo                       5    1652.814815     330.562963     15.05   <.0001 
 
 
Source                     DF    Type III SS    Mean Square   F Value   Pr > F 
 
combo                       5    1652.814815     330.562963     15.05   <.0001 
 
 
 Contrast                                   DF     Contrast SS     Mean Square 
 
 Main Effect for Temperature                 1     848.0740741     848.0740741 
 Main Effect for Bacteria                    2     651.8148148     325.9074074 
 Temperature by Bacteria Interaction         2     152.9259259      76.4629630 
 
           Contrast                               F Value    Pr > F 
 
           Main Effect for Temperature              38.61    <.0001 
           Main Effect for Bacteria                 14.84    <.0001 
           Temperature by Bacteria Interaction       3.48    0.0387 
  
_______________________________________________________________________________ 

                                Rotten potatoes                              14 
                    Two-factor ANOVA several different ways 
                                 Effect coding 
 
                               The REG Procedure 
                           Dependent Variable: rot  
 
                             Analysis of Variance 
  
                                    Sum of           Mean 
Source                   DF        Squares         Square    F Value    Pr > F 
 
Model                     5     1652.81481      330.56296      15.05    <.0001 
Error                    48     1054.22222       21.96296                      
Corrected Total          53     2707.03704                                     
 
 
             Root MSE              4.68647    R-Square     0.6106 
             Dependent Mean        9.40741    Adj R-Sq     0.5700 
 
                             Parameter Estimates 
  
                          Parameter       Standard 
     Variable     DF       Estimate          Error    t Value    Pr > |t| 
 
     Intercept     1        9.40741        0.63775      14.75      <.0001 
     b1            1       -4.12963        0.90191      -4.58      <.0001 
     b2            1       -0.24074        0.90191      -0.27      0.7907 
     t             1       -3.96296        0.63775      -6.21      <.0001 
     tb1           1        2.24074        0.90191       2.48      0.0165 
     tb2           1       -0.42593        0.90191      -0.47      0.6389 
_______________________________________________________________________________ 
 
              Test Temperature Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      848.07407      38.61    <.0001 
           Denominator        48       21.96296                      
  
 
               Test Bacteria Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2      325.90741      14.84    <.0001 
           Denominator        48       21.96296                      
 
 
             Test Bact_by_Temp Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2       76.46296       3.48    0.0387 
           Denominator        48       21.96296                      

_______________________________________________________________________________ 
  
                                Rotten potatoes                              18 
                    Two-factor ANOVA several different ways 
                  Further exploration using cell means coding 
 
Showing only the output from the test statements ... 
 
               Test Bact1vs2 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      136.11111       6.20    0.0163 
           Denominator        48       21.96296                      
  
               Test Bact1vs3 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      650.25000      29.61    <.0001 
           Denominator        48       21.96296                      
  
               Test Bact2vs3 Results for Dependent Variable rot 
 
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      191.36111       8.71    0.0049 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 
 
             Test Temp_for_Bac1 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1       53.38889       2.43    0.1255 
           Denominator        48       21.96296                      
  
 
             Test Temp_for_Bac2 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      346.72222      15.79    0.0002 
           Denominator        48       21.96296                      
 
             Test Temp_for_Bac3 Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      600.88889      27.36    <.0001 
           Denominator        48       21.96296                      

_______________________________________________________________________________ 
  
 
 
           Test Bact_for_CoolTemp Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2       47.44444       2.16    0.1264 
           Denominator        48       21.96296                      
  
 
           Test Bact_for_WarmTemp Results for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           2      354.92593      16.16    <.0001 
           Denominator        48       21.96296                      
  
_______________________________________________________________________________ 
 
                     Test Bact1vs2_for_WarmTemp Results 
                          for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      193.38889       8.81    0.0047 
           Denominator        48       21.96296                      
 
                      Test Bact1vs3_for_WarmTemp Results 
                          for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      709.38889      32.30    <.0001 
           Denominator        48       21.96296                      
  
 
                      Test Bact2vs3_for_WarmTemp Results 
                          for Dependent Variable rot 
  
                                           Mean 
           Source             DF         Square    F Value    Pr > F 
 
           Numerator           1      162.00000       7.38    0.0092 
           Denominator        48       21.96296                      

_______________________________________________________________________________ 
  
                                Rotten potatoes                              30 
                    Two-factor ANOVA several different ways 
            Table of critical values for all possible Scheffe tests 
 
        Initial test has         5  and         48 degrees of freedom. 
                  Using significance level alpha =       0.05 
 
 
        Number of Contrasts in followup test     Scheffe Critical Value 
 
                                           1                  12.042571 
                                           2                  6.0212853 
                                           3                  4.0141902 
                                           4                  3.0106426 
                                           5                  2.4085141 
 
 
 
SENIC data are unbalanced. 
 
/******************* senic2way.sas ****************/ 
%include 'senicreadxls.sas';  /* senicreadxls.sas reads data, etc.  */ 
title2 'Two-way ANCOVA  on SENIC Data'; 
 
proc glm; 
     class region mschool; 
     model infrisk = nbeds census nurses region|mschool; 
     lsmeans region|mschool / pdiff adjust=bon; 
 
/* Check relationships among explanatory variables */ 
 
proc freq; 
     tables mschool*region / nocol nopercent chisq; 
proc corr nosimple; 
     var nbeds census nurses; 
 

 
_______________________________________________________________________________ 
  
 
          Study of the Effectiveness of Nosocomial Infection Control          1 
                         Two-way ANCOVA  on SENIC Data 
 
                               The GLM Procedure 
 
                            Class Level Information 
  
          Class         Levels    Values 
 
          region             4    North Central Northeast South West  
 
          mschool            2    No Yes                              

                    Number of Observations Read         113 
                    Number of Observations Used         107 
  
_______________________________________________________________________________ 
  
          Study of the Effectiveness of Nosocomial Infection Control          2 
                         Two-way ANCOVA  on SENIC Data 
 
                               The GLM Procedure 
  
Dependent Variable: infrisk   Infection Risk 
 
                                      Sum of 
Source                     DF        Squares    Mean Square   F Value   Pr > F 
 
Model                      10     0.15630432     0.01563043      3.94   0.0002 
 
Error                      96     0.38078563     0.00396652                    
 
Corrected Total           106     0.53708994                                   
 
 
             R-Square     Coeff Var      Root MSE    infrisk Mean 
 
             0.291021      15.21644      0.062980        0.413896 
 
 
Source                     DF      Type I SS    Mean Square   F Value   Pr > F 
 
nbeds                       1     0.07821638     0.07821638     19.72   <.0001 
census                      1     0.01307734     0.01307734      3.30   0.0725 
nurses                      1     0.01244240     0.01244240      3.14   0.0797 
region                      3     0.02704967     0.00901656      2.27   0.0850 
mschool                     1     0.00186655     0.00186655      0.47   0.4944 
region*mschool              3     0.02365198     0.00788399      1.99   0.1210 
 
 
Source                     DF    Type III SS    Mean Square   F Value   Pr > F 
 
nbeds                       1     0.00126794     0.00126794      0.32   0.5731 
census                      1     0.00387959     0.00387959      0.98   0.3252 
nurses                      1     0.00683221     0.00683221      1.72   0.1925 
region                      3     0.00892874     0.00297625      0.75   0.5248 
mschool                     1     0.00238192     0.00238192      0.60   0.4403 
region*mschool              3     0.02365198     0.00788399      1.99   0.1210 

_______________________________________________________________________________ 
  
          Study of the Effectiveness of Nosocomial Infection Control          3 
                         Two-way ANCOVA  on SENIC Data 
 
                               The GLM Procedure 
                              Least Squares Means 
                Adjustment for Multiple Comparisons: Bonferroni 
 
                                         infrisk      LSMEAN 
                   region                 LSMEAN      Number 
 
                   North Central      0.40139751           1 
                   Northeast          0.43153167           2 
                   South              0.41847940           3 
                   West               0.39934025           4 
 
 
                    Least Squares Means for effect region 
                     Pr > |t| for H0: LSMean(i)=LSMean(j) 
                                        
                         Dependent Variable: infrisk 
  
         i/j              1             2             3             4 
 
            1                      1.0000        1.0000        1.0000 
            2        1.0000                      1.0000        1.0000 
            3        1.0000        1.0000                      1.0000 
            4        1.0000        1.0000        1.0000               
  
_______________________________________________________________________________ 
  
          Study of the Effectiveness of Nosocomial Infection Control          4 
                         Two-way ANCOVA  on SENIC Data 
 
                               The GLM Procedure 
                              Least Squares Means 
                Adjustment for Multiple Comparisons: Bonferroni 
 
                                               H0:LSMean1= 
                                    infrisk      LSMean2 
                    mschool          LSMEAN       Pr > |t| 
 
                    No           0.42165238         0.4403 
                    Yes          0.40372203                

_______________________________________________________________________________ 
  
          Study of the Effectiveness of Nosocomial Infection Control          5 
                         Two-way ANCOVA  on SENIC Data 
 
                               The GLM Procedure 
                              Least Squares Means 
                Adjustment for Multiple Comparisons: Bonferroni 
 
                                              infrisk      LSMEAN 
             region           mschool          LSMEAN      Number 
 
             North Central    No           0.42383291           1 
             North Central    Yes          0.37896211           2 
             Northeast        No           0.43749853           3 
             Northeast        Yes          0.42556481           4 
             South            No           0.38963669           5 
             South            Yes          0.44732210           6 
             West             No           0.43564139           7 
             West             Yes          0.36303911           8 
 
 
                Least Squares Means for effect region*mschool 
                     Pr > |t| for H0: LSMean(i)=LSMean(j) 
                                        
                         Dependent Variable: infrisk 
  
 i/j         1        2        3        4        5        6        7        8 
 
    1            1.0000   1.0000   1.0000   1.0000   1.0000   1.0000   1.0000 
    2   1.0000            1.0000   1.0000   1.0000   1.0000   1.0000   1.0000 
    3   1.0000   1.0000            1.0000   0.1938   1.0000   1.0000   1.0000 
    4   1.0000   1.0000   1.0000            1.0000   1.0000   1.0000   1.0000 
    5   1.0000   1.0000   0.1938   1.0000            1.0000   1.0000   1.0000 
    6   1.0000   1.0000   1.0000   1.0000   1.0000            1.0000   1.0000 
    7   1.0000   1.0000   1.0000   1.0000   1.0000   1.0000            1.0000 
    8   1.0000   1.0000   1.0000   1.0000   1.0000   1.0000   1.0000          
  
_______________________________________________________________________________ 

          Study of the Effectiveness of Nosocomial Infection Control          6 
                         Two-way ANCOVA  on SENIC Data 
 
                              The FREQ Procedure 
 
                          Table of mschool by region 
 
             mschool(Medical school affiliation) 
                       region(Region of country (usa)) 
 
             Frequency| 
             Row Pct  |Northeas|North Ce|South   |West    |  Total 
                      |t       |ntral   |        |        | 
             ---------+--------+--------+--------+--------+ 
             No       |     23 |     24 |     34 |     13 |     94 
                      |  24.47 |  25.53 |  36.17 |  13.83 | 
             ---------+--------+--------+--------+--------+ 
             Yes      |      5 |      7 |      3 |      2 |     17 
                      |  29.41 |  41.18 |  17.65 |  11.76 | 
             ---------+--------+--------+--------+--------+ 
             Total          28       31       37       15      111 
 
                             Frequency Missing = 2 
 
 
                   Statistics for Table of mschool by region 
 
            Statistic                     DF       Value      Prob 
            ------------------------------------------------------ 
            Chi-Square                     3      2.9284    0.4028 
            Likelihood Ratio Chi-Square    3      3.0488    0.3842 
            Mantel-Haenszel Chi-Square     1      1.0836    0.2979 
            Phi Coefficient                       0.1624           
            Contingency Coefficient               0.1603           
            Cramer's V                            0.1624           
 
             WARNING: 38% of the cells have expected counts less  
                      than 5. Chi-Square may not be a valid test. 
 
                          Effective Sample Size = 111 
                             Frequency Missing = 2 

          Study of the Effectiveness of Nosocomial Infection Control          7 
                         Two-way ANCOVA  on SENIC Data 
 
                              The CORR Procedure 
 
                   3  Variables:    nbeds    census   nurses    
 
 
                       Pearson Correlation Coefficients  
                          Prob > |r| under H0: Rho=0  
                            Number of Observations 
  
                                             nbeds        census        nurses 
 
 nbeds                                     1.00000       0.98100       0.91598 
 Average # beds during study period                       <.0001        <.0001 
                                               113           113           112 
 
 census                                    0.98100       1.00000       0.90989 
 Aver # patients in hospital per day        <.0001                      <.0001 
                                               113           113           112 
 
 nurses                                    0.91598       0.90989       1.00000 
 Aver # nurses during study period          <.0001        <.0001               
                                               112           112           112 
/dos/brunner/2101f12/lecture/senic >  
<.0001 


